Pure Functional Epidemics
An Agent-Based Approach

Jonathan Thaler Thorsten Altenkirch Peer-Olaf Siebers

University of Nottingham, United Kingdom

IFL 2018

TODOs

- reduce code as far as possible, only present susceptible
- add diagrams and explain FRP and arrowized notation on high level
- alternative story: sequentiality with RNG monad solves the problem and parallelism of Environment access ensures right semantics
- show video for 2d 21x21 grid until 195
- prob no future work slide, not enough space
- dont talk about "copies of Environments"

Research Question(s)

- **How** can we implement Agent-Based Simulation (ABS) in (pure) functional programming?
- **What** are the benefits and drawbacks?

Agent-Based Simulation (ABS)

Example

Simulate *the spread of an infectious disease in a city.*
What are the ***dynamics*** *(peak, duration of disease)?*

1. Start with population → Agents
2. Situated in City → Environment
3. Interacting with each other → local interactions
4. Creating dynamics → emergent system behaviour
5. Therefore ABS → bottom-up approach

SIR Model

- Population size $N = 1,000$
- Contact rate $\beta = 0.2$
- Infection probability $\gamma = 0.05$
- Illness duration $\delta = 15$
- 1 initially infected agent

System Dynamics

Top-Down, formalised using Differential Equations, give rise to dynamics.

SIR Model Dynamics



How-To implement ABS?

Established, state-of-the-art approach in ABS

Object-Oriented Programming in Python, Java,...

We want (pure) functional programming

Purity, explicit about side-effects, declarative, reasoning, parallelism, concurrency, property-based testing,...

How can we do it?

Functional Reactive Programming

Functional Reactive Programming (FRP)

- Continuous- & discrete-time systems in functional programming
- Signal Function (SF)
  $\Rightarrow$ process over time, maps signal to signal
- Events (deterministic & stochastic)
- Random-number streams
- *Arrowized* FRP using the *Yampa* library

First Steps

```haskell
data SIRState = Susceptible | Infected | Recovered

type SIRAgent = SF [SIRState] SIRState

sirAgent :: RandomGen g => g -> SIRState -> SIRAgent
sirAgent g Susceptible = susceptibleAgent g
sirAgent g Infected    = infectedAgent g
sirAgent _ Recovered   = recoveredAgent

recoveredAgent :: SIRAgent
recoveredAgent = arr (const Recovered)
```

Susceptible Agent

```haskell
susceptibleAgent :: RandomGen g => g -> SIRAgent
susceptibleAgent g = switch (susceptible g) (const (infectedAgent g))
  where
    susceptible :: RandomGen g => g -> SF [SIRState] (SIRState, Event ())
    susceptible g = proc as -> do
      makeContact <- occasionally g (1 / contactRate) () -< ()
      if isEvent makeContact
        then (do
          -- draw random element from the list
          a <- drawRandomElemSF g -< as
          case a of
            Infected -> do
              -- returns True with given probability
              i <- randomBoolSF g infectivity -< ()
              if i
                then returnA -< (Infected, Event ())
                else returnA -< (Susceptible, NoEvent)
            _        -> returnA -< (Susceptible, NoEvent))
        else returnA -< (Susceptible, NoEvent)
```

Infected Agent

```
1  infectedAgent :: RandomGen g => g -> SIRAgent
2  infectedAgent g = switch infected (const recoveredAgent)
3    where
4      infected :: SF [SIRState] (SIRState, Event ())
5      infected = proc _ -> do
6        recEvt <- occasionally g illnessDuration () -< ()
7        let a = event Infected (const Recovered) recEvt
8        returnA -< (a, recEvt)
```

Running the Simulation

```haskell
runSimulation :: RandomGen g => g -> Time -> DTime -> [SIRState] -> [[SIRState]]
runSimulation g t dt as = embed (stepSimulation sfs as) ((), dts)
  where
    steps     = floor (t / dt)
    dts       = replicate steps (dt, Nothing)
    n         = length as
    (rngs, _) = rngSplits g n [] -- unique rngs for each agent
    sfs       = zipWith sirAgent rngs as

stepSimulation :: [SIRAgent] -> [SIRState] -> SF () [SIRState]
stepSimulation sfs as =
    dpSwitch
      -- feeding the agent states to each SF
      (\_ sfs' -> (map (\sf -> (as, sf)) sfs'))
      -- the signal functions
      sfs
      -- switching event, ignored at t = 0
      (switchingEvt >>> notYet)
      -- recursively switch back into stepSimulation
      stepSimulation
  where
    switchingEvt :: SF ((), [SIRState]) (Event [SIRState])
    switchingEvt = arr (\ (_, newAs) -> Event newAs)
```

Dynamics $\Delta t = 0.1$



(a) Agent-Based approach



(b) System Dynamics

Dynamics $\Delta t = 0.01$



(a) Agent-Based approach



(b) System Dynamics

Reflection

- Time
- Agents
- Feedback
- Stochasticity
- Deterministic

Where is the Environment?

Adding an Environment

- Running SFs *'parallel'*
  $\Rightarrow$ multiple copies of environment
- State Monad elegant solution

Problem

Yampa not monadic

Solution

Monadic Stream Functions (MSFs)
$\Rightarrow$ Signal Functions with monadic context

The environment: Moore neighbourhood



Re-defining Types

```haskell
type Disc2dCoord = (Int, Int)
type SIREnv      = Array Disc2dCoord SIRState

type SIRMonad g = StateT SIREnv (Rand g)
type SIRAgent g = SF (SIRMonad g) () ()

neighbours :: SIREnv -> Disc2dCoord -> Disc2dCoord -> [Disc2dCoord] -> [SIRState]

moore :: [Disc2dCoord]
moore = [ topLeftDelta,    topDelta,     topRightDelta,
          leftDelta,                     rightDelta,
          bottomLeftDelta, bottomDelta,  bottomRightDelta ]

topLeftDelta :: Disc2dCoord
topLeftDelta      = (-1, -1)

topDelta :: Disc2dCoord
topDelta          = ( 0, -1)
...
```

Susceptible Agent revisited

```
1  susceptibleAgent :: RandomGen g => Disc2dCoord -> SIRAgent g
2  susceptibleAgent coord = switch susceptible (const (infectedAgent coord))
3    where
4      susceptible :: RandomGen g => SF (SIRMonad g) () ((), Event ())
5      susceptible = proc _ -> do
6        makeContact <- occasionallyM (1 / contactRate) () -< ()
7        if not (isEvent makeContact)
8          then returnA -< ((), NoEvent)
9          else (do
10           env <- arrM_ (lift get) -< ()
11           let ns = neighbours env coord agentGridSize moore
12           s <- drawRandomElemS -< ns
13           case s of
14             Infected -> do
15               infected <- arrM_
16                 (lift $ lift $ randomBoolM infectivity) -< ()
17               if infected
18                 then (do
19                   arrM (put . changeCell coord Infected) -< env
20                   returnA -< ((), Event ()))
21                 else returnA -< ((), NoEvent)
22             _          -> returnA -< ((), NoEvent))
```

Dynamics with environment



(a) Dynamics over $t = 200$



(b) Visualisation at $t = 100$

Conclusion

- Purity guarantees reproducibility at compile time
- Performance
- Agent-Identity a bit lost
- Agent-Interaction is main difficulty

Future Work

Property-based testing for verification in ABS

Might offer huge benefits e.g. formalising hypotheses

Dependent Types in ABS (using Idris)

- Safe environment access, state-transitions, agent-interactions
- Equilibrium of Model & Totality of Implementation

Thank You!